


```

' ----[ Title ]-----
' {$STAMP BS2}
' {$PBASIC 2.5}
' Voltmeter (8 bit)
' Revision 2
' Using ADC0831 at 5 volts
' Based on Parallax Samples
' Paul Ashley
' 3-18-2010
' ----[ Declarations ]-----
adcBits VAR Byte
v VAR Byte
r VAR Byte
g VAR Byte
v2 VAR Byte
v3 VAR Byte
' ----[ Initialization ]-----
CS PIN 0
CLK PIN 1
DataOutput PIN 2

DEBUG CLS
' ----[ Main Routine ]-----
DO
GOSUB ADC_Data
GOSUB Calc_Volts
GOSUB Display
LOOP
' ----[ Subroutines ]-----
ADC_Data:
HIGH CS
LOW CS
LOW CLK
PULSOUT CLK, 210
SHIFTIN DataOutput,CLK,MSBPOST,[adcBits]8]
RETURN


Calc_Volts:
g=adcBits      'Binary value to be converted
v = 5 * g / 255  'Multiply and divide
r = 5 * g // 255  '// returns the Modulus (remainder)
v2 = 100 * r / 255
v3 = 100 * r // 255
v3 = 10 * v3 / 255
IF (v3 >= 5) THEN v2 = v2 + 1
IF (v2 >= 100) THEN
v = v + 1
v2 = 0
ENDIF
RETURN
Display:
DEBUG HOME
DEBUG "8-bit binary value: ", BIN8 adcBits
DEBUG CR, CR, "Decimal value: ", DEC3 adcBits
DEBUG CR, CR, "DVM Reading: "
DEBUG DEC1 v, ".", DEC2 v2, " Volts"
RETURN

```



Quantity	Part
1	ADC0831 A to D converter -one channel
1	100K ohm Potentiometer
1	BS2 Microprocessor system

REVISIONS		



The potentiometer works as a voltage divider. It delivers 0 to 5 volts (Vcc) to Vin+ on the ADC. This results in binary values (0 to 255) sent from DO that represents the Vin+ value. The BS2 converts the binary to decimal and scales it to in input voltage.

This is a good circuit for testing ADC's for proper operation.

This demonstrates using an ADC0831 A to D converter to measure a voltage divider in the circuit. The results is displayed using the BS2 debug screen. This can be the basis for measuring several types of sensors that are based on a voltage divider. The ADC0831 is an 8 bit converter that gives a value that is between 0 and 255.

TITLE		
Voltmeter using an ADC For the Parallax BS2		
DATE 3-18-2010	SCALE none	Parallax BS2 program provided.
DRAWN BY Paul Ashley	PAGES 1 of 1	
		www.robo-works.net